Parameterization choices for hierarchical models: centered, non-centered

Overview

This note expands on the section in Bob Carpenter's most excellent case study Hierarchical Partial Pooling for Repeated Binary Trials which introduces the non-centered parameterization of a hierarchical model.

Hierarchical models provide partial pooling of information across parameters according to group membership. The hierarchical model provides group-level parameters which influence the fixed-effects parameters (individual distributions on the group members). For datasets where the groups have relatively few members, MCMC samplers cannot easily explore the resulting sampling density. The non-centered parameterization mitigates this problem by decoupling the group-level and fixed-effects parameters in the sampling distribution.

The case study dataset is taken from baseball, consisting of the number of hits and at-bats for a set of Major League Baseball players. While individual players have differing batting abilities, they are taken from the population of MLB baseball players. Therefore, it makes sense to build a hierarchical model of player ability.

The data consists of N observations y , where each observation y_n is the number of successes for $player_n$ in K trials. The dataset is small: 18 players (N=18), 45 at-bats (K=45). The model estimates θ_n , each player's chance of success for an at-bat. $(\theta*1000$ is a player's "batting average".) It does so by recasting the problem in terms of parameter α , a player's log-odds of success. The hierarchical model puts a normal prior with group-level parameters μ and σ on the estimates for parameter α , which pulls the individual player estimates towards the group mean μ .

The log-odds parameterization makes it much easier to expand the model by adding more fixed effects and other multilevel effects. The change of success theta is computed in the model's generated quantities block as inv_logit(alpha).

The centered parameterization for a hierarchical model corresponds directly to the data structure: the individual-level parameter alpha - a player's log-odds of success is given a prior distribution specified in terms of the group-level parameters: alpha ~ normal(mu, sigma). The non-centered parameterization is recommended for hierarchical models where the groups have relatively few members. The trick is to decouple alpha, mu, and sigma in the sampling distribution by reparameterization. There are two ways to do this reparameterization:

- nere are two ways to do this reparameterization.
- a non-centered parameterization with standard normal prior on parameter alpha_std and auxiliary variable alpha.
- a non-centered parameterization using an affine transform on parameter alpha.

In this note we show models for each and then plot the result.

Packages used in this notebook

We use CmdStanPy to do the model fitting and plot the results using plotnine, a ggplot2-like Python package. Pandas and NumPy are also used for data munging.

Baseball Data: Number of hits in 45 at-bats for 18 MLB players in 1971

Out[3]:	FirstName	LastName	At- Bats	Hits	BattingAverage	RemainingAt- Bats	RemainingAverage	SeasonAt- Bats	SeasonHits
	Roberto	Clemente	45	18	0.400	367	0.346	412	145
	Frank	Robinson	45	17	0.378	426	0.298	471	144
	Frank	Howard	45	16	0.356	521	0.276	566	160
	Jay	Johnstone	45	15	0.333	275	0.222	320	76
	Ken	Berry	45	14	0.311	418	0.273	463	128
	Jim	Spencer	45	14	0.311	466	0.270	511	140
	Don	Kessinger	45	13	0.289	586	0.265	631	168
	Luis	Alvarado	45	12	0.267	138	0.210	183	41
	Ron	Santo	45	11	0.244	510	0.269	555	148
	Ron	Swaboda	45	11	0.244	200	0.230	245	57
	Rico	Petrocelli	45	10	0.222	538	0.264	583	152
	Ellie	Rodriguez	45	10	0.222	186	0.226	231	52
	George	Scott	45	10	0.222	435	0.303	480	142
	Del	Unser	45	10	0.222	277	0.264	322	83
	Billy	Williams	45	10	0.222	591	0.330	636	205
	Bert	Campaneris	45	9	0.200	558	0.285	603	168
	Thurman	Munson	45	8	0.178	408	0.316	453	137
	Max	Alvis	45	7	0.156	70	0.200	115	21

```
M = 10000 # desired number of draws from the posterior

# ggplot2 x_y plot with axis labels and optional title
def scatter_plot(df, x_lab, y_lab, title=''):
    return (ggplot(df, aes('x', 'y')) +
        geom_point(alpha=0.2) +
        xlab(x_lab) +
        ylab(y_lab) +
        ggtitle(title) +
        theme(figure_size=(8,6)))
```

The Model

The model we are interested in is a hierarchical model with a *normal prior* on the *log odds of success*. The mathematical model specification is

$$p(y_n | K_n, \alpha_n) = \mathsf{Binomial}(y_n | K_n, \mathsf{logit}^{-1}(\alpha_n))$$

with a simple normal hierarchical prior

$$p(\alpha_n \mid \mu, \sigma) = \mathsf{Normal}(\alpha_n \mid \mu, \sigma).$$

a weakly informative hyperprior for μ

$$p(\mu) = \mathsf{Normal}(\mu \mid -1, 1),$$

and a half normal prior on σ

$$p(\sigma) = 2 \operatorname{\mathsf{Normal}}(\sigma \,|\, 0,1) \propto \operatorname{\mathsf{Normal}}(\sigma \,|\, 0,1).$$

Centered Parameterization

The Stan program hier-logit-centered.stan is a straightforward encoding of a hierarchical model with a normal prior on the log odds of success, but this is not the optimal way to code this model in Stan, as we will soon demonstrate.

The chance of success θ is computed as a generated quantity.

```
generated quantities {
  vector[N] theta = inv_logit(alpha);
}
```

In CmdStanPy, model fitting is done in two steps: first instantiate the model object from a Stan program file; then run the Stan inference algorithm, here the NUTS-HMC sampler, which returns the inferences.

We instantiate the CmdStanModel object from the Stan program file 'hier-logit-centered.stan'. By default, CmdStanPy compiles the model on object instantiation, unless there is a corresponding exe file which has a more recent timestamp than the source file. The model's code method returns the Stan program.

```
In [5]:
         hier_logit_centered_model = CmdStanModel(stan_file='hier-logit-centered.stan')
         print(hier logit centered model.code())
        INFO:cmdstanpy:found newer exe file, not recompiling
        data {
          int<lower=0> N; // items
          array[N] int<lower=0> K; // initial trials
          array[N] int<lower=0> y; // initial successes
        }
        parameters {
          real mu; // population mean of success log-odds
          real<lower=0> sigma; // population sd of success log-odds
          vector[N] alpha; // success log-odds
        }
        model {
          mu ~ normal(-1, 1); // hyperprior
          sigma ~ normal(0, 1); // hyperprior
          alpha ~ normal(mu, sigma); // prior (hierarchical)
          y ~ binomial_logit(K, alpha); // likelihood
        generated quantities {
          vector[N] theta = inv logit(alpha);
        }
```

Next run the NUTS-HMC sampler. By default the sampler runs 4 chains. The argument <code>iter_sampling</code> specifies the *per-chain* number of sampling iterations. The defaults are 1000 warmup and 1000 sampling iterations per chain, for a sample containing a total of 4000 draws. Since M=10000, we override this default. We specify the random seed for reproducibility.

INFO:cmdstanpy:CmdStan start processing

INFO:cmdstanpy:CmdStan done processing.

The variable theta is the per-player chance of success, i.e., theta * 1000 is a player's batting average. The estimates range from 0.24 to 0.3, batting averages between 240 and a respectable 300, which is in line with what we know about major league baseball players.

```
fit_centered.summary(sig_figs=3).round(decimals=3).filter(
In [7]:
              regex=r'mu|sigma|theta', axis="index")
                   Mean MCSE StdDev
                                          5%
                                                50%
                                                       95%
                                                              N_Eff N_Eff/s R_hat
Out[7]:
            name
              mu -1.030 0.002
                                  0.094 -1.180 -1.030 -0.873 2230.0
                                                                    3400.0
                                                                             1.00
            sigma 0.189
                          0.004
                                  0.103 0.054
                                                0.173
                                                     0.382
                                                              696.0
                                                                    1060.0
                                                                             1.01
          theta[1] 0.298
                          0.001
                                  0.045 0.239
                                               0.291 0.382 2188.0
                                                                     3325.0
                                                                             1.00
          theta[2]
                   0.292
                          0.001
                                  0.041
                                        0.237
                                               0.286
                                                      0.370
                                                             2826.0
                                                                     4294.0
                                                                             1.00
                                                                             1.00
          theta[3] 0.287
                          0.001
                                  0.039 0.233 0.282 0.360 3337.0
                                                                    5071.0
```

	Mean	MCSE	StdDev	5%	50%	95%	N_Eff	N_Eff/s	R_hat
name									
theta[4]	0.282	0.001	0.038	0.228	0.278	0.350	4080.0	6201.0	1.00
theta[5]	0.277	0.001	0.036	0.224	0.273	0.342	4967.0	7549.0	1.00
theta[6]	0.277	0.001	0.037	0.222	0.273	0.344	5192.0	7891.0	1.00
theta[7]	0.271	0.000	0.036	0.216	0.269	0.332	6397.0	9723.0	1.00
theta[8]	0.266	0.000	0.034	0.211	0.265	0.324	5932.0	9015.0	1.00
theta[9]	0.261	0.000	0.035	0.204	0.260	0.318	5166.0	7851.0	1.00
theta[10]	0.260	0.000	0.034	0.204	0.261	0.316	4744.0	7210.0	1.00
theta[11]	0.255	0.001	0.034	0.198	0.256	0.310	4289.0	6519.0	1.00
theta[12]	0.255	0.001	0.034	0.197	0.257	0.310	4539.0	6898.0	1.00
theta[13]	0.255	0.001	0.035	0.195	0.256	0.310	3886.0	5906.0	1.00
theta[14]	0.255	0.001	0.035	0.196	0.256	0.310	4018.0	6106.0	1.00
theta[15]	0.255	0.001	0.034	0.198	0.256	0.309	3790.0	5760.0	1.00
theta[16]	0.250	0.001	0.035	0.189	0.252	0.304	3067.0	4660.0	1.00
theta[17]	0.245	0.001	0.036	0.181	0.248	0.299	2293.0	3484.0	1.00
theta[18]	0.241	0.001	0.037	0.174	0.244	0.296	1877.0	2853.0	1.00

E00/

0.5%

N Eff N Eff/c D bot

The reported Eff values for sigma are low and the R_hat value is above 1. CmdStan's diagnose method indicates that this model had problems fitting the data.

```
In [8]: print(fit_centered.diagnose())
```

Processing csv files: /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-centered-20220129122528_1.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-centered-20220129122528_2.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-centered-20220129122528_3.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-centered-20220129122528_3.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-centered-20220129122528_4.csv

Checking sampler transitions treedepth. Treedepth satisfactory for all transitions.

Moon MCCE StdDov

Checking sampler transitions for divergences.

76 of 10000 (0.76%) transitions ended with a divergence.

These divergent transitions indicate that HMC is not fully able to explore the posterior ${\tt d}$ is tribution.

Try increasing adapt delta closer to 1.

If this doesn't remove all divergences, try to reparameterize the model.

Checking E-BFMI - sampler transitions HMC potential energy.

The E-BFMI, 0.22, is below the nominal threshold of 0.3 which suggests that HMC may have t rouble exploring the target distribution.

If possible, try to reparameterize the model.

Effective sample size satisfactory.

Split R-hat values satisfactory all parameters.

Processing complete.


The Funnel

These diagnostics indicate that the sampler failed to fit the data and that the resulting sample is not a sample from the true posterior. The reason for this failure is that given the small amount of data, the

sampler cannot properly determine how much of the observed variance in the data is individual-level variance, parameter alpha, or group-level variance, parameter sigma. The diagnostics report low ESS and poor R-hat for sigma.

Plotting the estimate of alpha[1], the log-odds success for player 1, against log(sigma), the group-level variance, provide additional evidence of the problem. This plot shows a clear funnel shape with many draws at the bottom of the neck of the funnel. This is the reason for the low EFF numbers for sigma. The sampler "gets stuck" at the bottom of the funnel. The algorithm tries to jump to a new point, but large jumps fall outside of the posterior density, resulting in a divergence. Small jumps fail to exit the neck of the funnel.

hierarchical vs fixed, centered parameterization



Out[9]: <ggplot: (8773807874714)>

The Non-Centered Parameterization

Instead of a hierarchical prior, the non-centered parameterization takes a standard unit normal prior for a new variable,

$$lpha_n^{
m std} = rac{lpha_n - \mu}{\sigma}.$$

Then we can parameterize in terms of $lpha^{
m std}$, which has a standard-normal distribution

$$p(\alpha_n^{ ext{std}}) = \mathsf{Normal}(\alpha_n^{ ext{std}} \,|\, 0, 1).$$

We can then define our original α as a derived quantity.

$$\alpha_n = \mu + \sigma \, \alpha_n^{\rm std}$$
.

This decouples the sampling distribution for $\alpha^{\rm std}$ from μ and σ , greatly reducing their correlation in the posterior. The sampler only knows about the model parameters. Since the prior on parameter α is not specified in terms of parameters μ and σ , the sampler can move more freely along their axes, and therefore explore the posterior more fully. Although we decouple the parameters, we still need to share information between the group-level and individual level parameters; this is done using auxiliary variables, either transformed parameters or directly in the model block.

Non-centered parameterization using a standard normal distribution

Prior to Stan 2.19, a Stan implementation directly encoded the above reparameterization. This requires 3 changes to the centered parameterization:

- In the parameters block, declaring a parameter alpha_std (instead of parameter alpha). This name implies that it will have a standard normal distribution.
- ullet In the transformed parameters block define variable alpha as mu + sigma * alpha_std .
- In the model block we put a standard normal prior on alpha_std , which decouples the sampling distribution of alpha_std from mu and sigma .

The Stan program "hier-logit-nc-std-norm.stan" follows this pattern.

```
data {
  int<lower=0> N; // items
  array[N] int<lower=0> K; // initial trials
  array[N] int<lower=0> y; // initial successes
}
parameters {
  real mu; // population mean of success log-odds
  real<lower=0> sigma; // population sd of success log-odds
  vector[N] alpha_std; // success log-odds (standardized)
}
transformed parameters {
  vector[N] alpha = mu + sigma * alpha_std;
}
model {
  mu \sim normal(-1, 1); // hyperprior
  sigma ~ normal(0, 1); // hyperprior
  alpha_std ~ normal(0, 1); // prior (hierarchical)
  y ~ binomial_logit(K, alpha); // likelihood
generated quantities {
```

```
vector[N] theta = inv_logit(alpha);
}
```

Non-centered parameterization using an affine transform

Since Stan version 2.19, the Stan language's affine transform construct provides a more concise way to do this. For a real variable, the affine transform $x\mapsto \mu+\sigma*x$ with offset μ and (positive) multiplier σ is specified using a syntax like that used for upper/lower bounds, with keywords offset, multiplier. Specifying the affine transform in the parameter declaration for $\alpha^{\rm std}$ eliminates the need for intermediate variables and makes it easier to see the hierarchical structure of the model.

When the parameters to the prior for σ are constants, the normalization for the half-prior (compared to the full prior) is constant and therefore does not need to be included in the notation. This only works if the parameters to the density are data or constants; if they are defined as parameters or as quantities depending on parameters, then explicit truncation is required.

The Stan program hier-logit-nc-affine-xform.stan uses the affine-transform syntax to specify the non-centered version of the hierarchical model with a normal prior on the log odds of success.

```
data {
  int<lower=0> N; // items
  array[N] int<lower=0> K; // initial trials
  array[N] int<lower=0> y; // initial successes
}
parameters {
  real mu; // population mean of success log-odds
  real<lower=0> sigma; // population sd of success log-odds
  vector<offset=mu, multiplier=sigma>[N] alpha; // success log-odds
(standardized)
}
model {
  mu \sim normal(-1, 1); // hyperprior
  sigma ~ normal(0, 1); // hyperprior
  alpha ~ normal(mu, sigma); // prior (hierarchical)
  y ~ binomial_logit(K, alpha); // likelihood
}
generated quantities {
  vector[N] theta = inv_logit(alpha);
  vector[N] alpha_std = (alpha - mu)/sigma;
}
```

Fitting the standard normal reparameterization

The model hier-logit-nc-std-norm.stan fits the model using parameter alpha_std.

Full disclosure: the choice of random seed '54321' was far from random; this seed allows the sampler to fit the model without divergences. Other seeds may result in 1 or 2 divergences for a sample of 2500 draws.

```
INFO:cmdstanpy:CmdStan start processing
data {
  int<lower=0> N; // items
  array[N] int<lower=0> K; // initial trials
  array[N] int<lower=0> y; // initial successes
parameters {
  real mu; // population mean of success log-odds
  real<lower=0> sigma; // population sd of success log-odds
  vector[N] alpha std; // success log-odds (standardized)
transformed parameters {
  vector[N] alpha = mu + sigma * alpha std;
}
model {
  mu ~ normal(-1, 1); // hyperprior
  sigma ~ normal(0, 1); // hyperprior
  alpha_std ~ normal(0, 1); // prior (hierarchical)
  y ~ binomial logit(K, alpha); // likelihood
}
generated quantities {
  vector[N] theta = inv logit(alpha);
```

INFO:cmdstanpy:CmdStan done processing.

Again, we check for problems by running CmdStan's diagnose method.

```
In [11]: print(fit_nc_std_norm.diagnose())
```

Processing csv files: /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-std-norm-20220129122529_1.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-std-norm-20220129122529_2.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-std-norm-20220129122529_3.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-std-norm-20220129122529_4.csv

Checking sampler transitions treedepth. Treedepth satisfactory for all transitions.

Checking sampler transitions for divergences. No divergent transitions found.

Checking E-BFMI - sampler transitions HMC potential energy. E-BFMI satisfactory.

Effective sample size satisfactory.

Split R-hat values satisfactory all parameters.

Processing complete, no problems detected.


The estimates for mu, sigma, theta and alpha are roughly the same as for the centered parameterization. The non-centered parameterization results in a much larger effective sample size.

```
"theta[1]", "theta[5]", "theta[10]", "theta[18]",
    "alpha[1]", "alpha[5]", "alpha[10]", "alpha[18]"],
    axis="index"))
```

```
Centered parameterization
           Mean MCSE StdDev
                                 5%
                                       50%
                                             95% N Eff N Eff/s R hat
name
         -1.030 0.002
                        0.094 -1.180 -1.030 -0.873 2230.0
                                                          3400.0
                                                                   1.00
sigma
          0.189
                0.004
                        0.103 0.054 0.173 0.382
                                                  696.0
                                                          1060.0
                                                                   1.01
          0.298 0.001
                        0.045 0.239 0.291 0.382 2188.0
                                                          3325.0
                                                                   1.00
theta[1]
                        0.036 0.224 0.273 0.342 4967.0
theta[5]
          0.277 0.001
                                                          7549.0
                                                                   1.00
theta[10] 0.260 0.000
                        0.034 0.204 0.261 0.316 4744.0
                                                          7210.0
                                                                   1.00
theta[18] 0.241 0.001 0.037 0.174 0.244 0.296 1877.0
                                                          2853.0
                                                                  1.00
alpha[1] -0.863 0.004 0.209 -1.160 -0.892 -0.480 2223.0
                                                          3378.0
                                                                   1.00
alpha[5] -0.968 0.003
                        0.180 -1.240 -0.978 -0.654 5031.0
                                                          7645.0
                                                                   1.00
alpha[10] -1.050 0.003 0.181 -1.360 -1.040 -0.770 4591.0
                                                          6978.0
                                                                   1.00
alpha[18] -1.160 0.005 0.211 -1.560 -1.130 -0.867 1911.0
                                                          2905.0
                                                                  1.00
Non-centered parameterization, std normal reparameterization
           Mean
                MCSE StdDev
                                 5%
                                       50%
                                             95%
                                                   N Eff N Eff/s R hat
name
         -1.030 0.001
                        0.090 -1.170 -1.030 -0.878
mu
                                                   8040.0 12300.0
                                                                    1.0
                0.002
                        0.110 0.017 0.148 0.364
sigma
          0.164
                                                   4120.0
                                                           6280.0
                                                                    1.0
          0.293 0.001
                        0.043 0.240 0.284 0.378
theta[1]
                                                   6968.0
                                                          10622.0
                                                                    1.0
          0.274 0.000
                       0.034 0.225 0.271 0.336
theta[5]
                                                  9617.0 14660.0
                                                                    1.0
theta[10] 0.261 0.000
                       0.032 0.208 0.262 0.314 10816.0 16487.0
                                                                    1.0
theta[18] 0.244 0.000 0.036 0.178 0.249 0.295
                                                  7175.0 10938.0
                                                                    1.0
alpha[1] -0.887 0.002 0.202 -1.150 -0.924 -0.496 6982.0 10643.0
                                                                    1.0
alpha[5] -0.978 0.002 0.170 -1.240 -0.989 -0.681 9709.0 14801.0
                                                                    1.0
alpha[10] -1.050 0.002 0.170 -1.340 -1.040 -0.779 10581.0 16130.0
                                                                    1.0
alpha[18] -1.140 0.002 0.207 -1.530 -1.100 -0.874
                                                 7012.0 10689.0
                                                                    1.0
```

To consider how the reparameterization is working, we plot the posterior for the mean and log scale of the hyperprior. The prior location (μ) and scale (σ) are coupled in the posterior.


Hierarchical params, standard normal reparameterization



Out[13]: <ggplot: (8773807914879)>

Now when we plot the sample values for log scale and the first transformed parameter, alpha_std[1], the range of both the X and Y axis are much wider. There is a diffuse set of points in the bottom half of the plot, not many points at the bottom of the Y axis. This indicates that the sampler has been able to properly explore the posterior density and therefore we have a valid sample from the posterior. As log sigma approaches zero the plot has a long right-hand tail.

hierarchical vs fixed param, non-centered parameterization




alpha_std[1]: player 1 log odds of success (transformed)

Out[14]: <ggplot: (8773771343047)>

We can also plot the value for the generated quantities variable alpha[1] against log(sigma) and compare it to the first plot from the centered parameterization. We recover the funnel shape, but now the Y axis ranges from (-12, 0) instead of (-4, 0). As above, as log sigma approaches zero the plot has a long right-hand tail.

hierarchical param vs generated quantity variable



Out[15]: <ggplot: (8773771132164)>

We still don't have enough data to determine whether or not the observed variance is hierarchical or individual-level variance. The model still provides us with an estimate for alpha, a player's log-odds of success at bat. Critically, because alpha is no longer a parameter variable, replaced by alpha_std in the sampling distribution, the sampler can fully explore the posterior.

Fitting the affine transform parameterization

The model hier-logit-nc-affine-xform.stan looks just like the centered parameterization, with the exception that parameter alpha is defined with <offset = mu, multiplier = sigma>.

To show that the affine transform reparameterization and the standard normal reparameterization are equivalent we fit the model to the data and plot the results.

```
In [16]: nc_affine_xform_model = CmdStanModel(stan_file='hier-logit-nc-affine-xform.stan')
    print(nc_affine_xform_model.code())

INFO:cmdstanpy:found newer exe file, not recompiling
    data {
        int<lower=0> N; // items
        array[N] int<lower=0> K; // initial trials
        array[N] int<lower=0> y; // initial successes
    }
    parameters {
        real mu; // population mean of success log-odds
        real<lower=0> sigma; // population sd of success log-odds
        vector<offset=mu, multiplier=sigma>[N] alpha; // success log-odds (standardized)
}
```

```
model {
           mu ~ normal(-1, 1); // hyperprior
           sigma ~ normal(0, 1); // hyperprior
           alpha ~ normal(mu, sigma); // prior (hierarchical)
           y ~ binomial logit(K, alpha); // likelihood
         generated quantities {
           vector[N] theta = inv logit(alpha);
           vector[N] alpha_std = (alpha - mu)/sigma;
          fit nc affine = nc affine xform model.sample(
In [17]:
              data=baseball data,
              iter sampling=int(M/4),
              seed=54321)
         INFO:cmdstanpy:CmdStan start processing
         INFO:cmdstanpy:CmdStan done processing.
         As usual, we check for problems by running CmdStan's diagnose method.
In [18]: | print(fit_nc_affine.diagnose())
         Processing csv files: /var/folders/db/4jnggnf549s42z50bd61jskm0000gg/T/tmphbww0ou /hier-lo
```

```
Processing csv files: /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-lo git-nc-affine-xform-20220129122533_1.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-affine-xform-20220129122533_2.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-affine-xform-20220129122533_3.csv, /var/folders/db/4jnggnf549s42z50bd61jskm0000gq/T/tmphbww0ou_/hier-logit-nc-affine-xform-20220129122533_4.csv

Checking sampler transitions treedepth.
Treedepth satisfactory for all transitions.

Checking sampler transitions for divergences.
No divergent transitions found.

Checking E-BFMI - sampler transitions HMC potential energy.
E-BFMI satisfactory.

Effective sample size satisfactory all parameters.

Processing complete, no problems detected.
```

```
print("Centered parameterization")
In [19]:
          print(fit centered.summary(sig figs=3).round(decimals=3).filter(
              ["mu", "sigma",
               "theta[1]", "theta[5]", "theta[10]", "theta[18]",
               "alpha[1]", "alpha[5]", "alpha[10]", "alpha[18]"],
              axis="index"))
          print("\nNon-centered parameterization, std normal reparameterization")
          print(fit_nc_std_norm.summary(sig_figs=3).round(decimals=3).filter(
              ["mu", "sigma",
               "theta[1]", "theta[5]", "theta[10]", "theta[18]",
               "alpha[1]", "alpha[5]", "alpha[10]", "alpha[18]"],
              axis="index"))
          print("\nNon-centered parameterization, affine transform reparameterization")
          print(fit nc affine.summary(sig figs=3).round(decimals=3).filter(
              ["mu", "sigma",
```


```
"theta[1]", "theta[5]", "theta[10]", "theta[18]",
    "alpha[1]", "alpha[5]", "alpha[10]", "alpha[18]"],
    axis="index"))
```

Centered parameterization

```
50%
                                              95%
                                                   N Eff N Eff/s R hat
           Mean MCSE StdDev
                                  5%
name
         -1.030 0.002
                        0.094 -1.180 -1.030 -0.873 2230.0
                                                            3400.0
                                                                     1.00
sigma
          0.189
                0.004
                        0.103 0.054 0.173 0.382
                                                    696.0
                                                            1060.0
                                                                     1.01
                        0.045 0.239 0.291 0.382 2188.0
          0.298 0.001
                                                            3325.0
                                                                     1.00
theta[1]
          0.277
                0.001
                        0.036 0.224 0.273 0.342
                                                   4967.0
                                                            7549.0
                                                                     1.00
theta[5]
theta[10] 0.260 0.000
                        0.034 0.204
                                     0.261 0.316
                                                   4744.0
                                                            7210.0
                                                                     1.00
theta[18] 0.241 0.001
                        0.037 0.174 0.244 0.296 1877.0
                                                            2853.0
                                                                     1.00
alpha[1] -0.863 0.004
                        0.209 -1.160 -0.892 -0.480 2223.0
                                                            3378.0
                                                                     1.00
alpha[5] -0.968 0.003
                        0.180 -1.240 -0.978 -0.654 5031.0
                                                            7645.0
                                                                     1.00
alpha[10] -1.050 0.003
                        0.181 -1.360 -1.040 -0.770 4591.0
                                                            6978.0
                                                                     1.00
alpha[18] -1.160 0.005
                        0.211 -1.560 -1.130 -0.867 1911.0
                                                            2905.0
                                                                     1.00
Non-centered parameterization, std normal reparameterization
                 MCSE StdDev
                                  5%
                                        50%
                                              95%
                                                     N Eff N Eff/s R hat
           Mean
name
mu
         -1.030 0.001
                        0.090 -1.170 -1.030 -0.878
                                                    8040.0 12300.0
                                                                       1.0
                0.002
                                     0.148 0.364
sigma
          0.164
                        0.110 0.017
                                                    4120.0
                                                            6280.0
                                                                       1.0
theta[1]
          0.293
                0.001
                        0.043
                               0.240
                                     0.284 0.378
                                                    6968.0
                                                            10622.0
                                                                      1.0
theta[5]
          0.274
                0.000
                        0.034
                               0.225
                                     0.271
                                            0.336
                                                    9617.0
                                                            14660.0
                                                                      1.0
theta[10] 0.261 0.000
                        0.032 0.208 0.262 0.314
                                                   10816.0
                                                            16487.0
                                                                       1.0
theta[18] 0.244 0.000
                        0.036 0.178 0.249 0.295
                                                    7175.0 10938.0
                                                                      1.0
alpha[1] -0.887 0.002
                                                    6982.0 10643.0
                        0.202 - 1.150 - 0.924 - 0.496
                                                                      1.0
alpha[5] -0.978 0.002
                        0.170 -1.240 -0.989 -0.681
                                                    9709.0 14801.0
                                                                      1.0
alpha[10] -1.050 0.002
                        0.170 -1.340 -1.040 -0.779 10581.0 16130.0
                                                                      1.0
alpha[18] -1.140 0.002 0.207 -1.530 -1.100 -0.874
                                                   7012.0 10689.0
                                                                      1.0
Non-centered parameterization, affine transform reparameterization
                 MCSE StdDev
           Mean
                                  5%
                                        50%
                                              95%
                                                     N Eff N Eff/s R hat
name
         -1.030 0.001
                        0.092 -1.180 -1.030 -0.882 10600.0 16000.0
mu
                                                                       1.0
          0.165 0.002
                        0.112 0.016 0.149 0.369
                                                    3870.0
                                                            5850.0
sigma
                                                                       1.0
          0.293 0.000
                        0.042 0.241
                                     0.284 0.376
                                                    7682.0
                                                            11605.0
                                                                       1.0
theta[1]
theta[5]
          0.274
                0.000
                        0.034
                               0.225
                                     0.271
                                            0.336
                                                   13773.0
                                                            20805.0
                                                                       1.0
         0.261
                                     0.261
                                            0.313
                                                   13102.0
theta[10]
                0.000
                        0.033
                               0.207
                                                            19792.0
                                                                       1.0
theta[18] 0.243
                0.000
                        0.036 0.178
                                     0.249 0.293
                                                    7881.0
                                                            11905.0
                        0.198 -1.150 -0.923 -0.508
alpha[1] -0.887
                0.002
                                                    7695.0
                                                            11624.0
                                                                       1.0
alpha[5]
         -0.979
                0.001
                        0.171 - 1.240 - 0.990 - 0.681
                                                   13781.0
                                                            20817.0
                                                                      1.0
alpha[10] -1.050 0.002
                        0.172 -1.340 -1.040 -0.785
                                                   12765.0 19282.0
                                                                      1.0
                        0.205 -1.530 -1.100 -0.880
                                                    7619.0 11509.0
alpha[18] -1.140 0.002
                                                                       1.0
```

We plot the sample values for log scale and the first player ability parameter, alpha[1]. This plot is almost identical to the above plot, "hierarchical vs generated quantity variable". Critically, this plot differs from the first plot from the centered parameterization.

hierarchical vs fixed param, affine transform



Out[20]: <ggplot: (8773808223711)>

Don't Panic!


You may be asking: "but this is a funnel plot, isn't this bad?"

The answer is: "no!

Stan reports the parameter estimates on the *constrained* scale, but it computes on the *unconstrained* scale. The corresponding unconstrained value for alpha is $\frac{\alpha_n - \mu}{\sigma}$. In the generated quantities block we recover this as variable alpha_std.

Plotting log(sigma) against alpha_std[1] we see the same sampling distribution as in the standard normal parameterization.

hierarchical vs fixed param, sampling distribution



Out[21]: <ggplot: (8773788123534)>

Both hier-logit-nc-std-norm.stan and hier-logit-nc-affine-xform.stan produce essentially the same results; this is because both models are essentially the same model: they encode the non-centered parameterization.

In program hier-logit-nc-std-norm.stan we define alpha as a transformed parameter and recover theta in the generated quantities block.

```
transformed parameters {
  vector[N] alpha = mu + sigma * alpha_std;
}
...
generated quantities {
  vector[N] theta = inv_logit(mu + sigma * alpha_std);
}
```

In program hier-logit-nc-affine-xform.stan variable alpha is a parameter with hierarchical prior normal(mu, sigma). In the generated quantities block we recover theta, our estimate of a player's chance of success. Were there a need for it, we would be able to generate variable alpha_std as well.

```
generated quantities {
  vector[N] theta = inv_logit(alpha);
  vector[N] alpha_std = (alpha - mu)/sigma;
}
```

Discussion

Hierarchical models where the of hierarchical prior is specified in terms of a location and scale can be parameterized in one of two ways: centered or non-centered. When there are enough per-group observations, the sampler can determine the amount of group-level variance from the amount of individual-level variance and the centered parameterization is recommended. For smaller amounts of per-group observations, the non-centered parameterization is preferred.

For the non-centered parameterization, using the affine transform makes it easier to see the hierarchical structure of the model. When using the affine transform, the sampler computes on the unconstrained scale reports the parameter value on the constrained scale. For this reason, using standard normal parameterization may be more computationally efficient, as it eleminates extra transforms, but for simple models this difference may not be noticeable.

In this note we only consider models with a normal hierarchical prior, which can be coded either by use of the Stan language's offset, multiplier syntax, or by explicitly introducing a standardized parameter. Non-normal hierarchical priors are more challenging to reparameterize and are beyond the scope of this discussion.